Math 174
Calculus II

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 21

Parametric Curves, Tangents and Areas

Initializations

```
> restart;
  with(plots):
```

21.1 Tangent Lines and Areas

The equation of the tangent line at the point $P = (x(t_0), y(t_0))$ to the curve

$$x = x(t) \qquad y = y(t)$$

is given by

$$y - y(t_0) = \frac{y'(t_0)}{x'(t_0)}(x - x(t_0))$$

provided of course that $x'(t_0) \neq 0$. If $x'(t_0) = 0$, then the tangent line is vertical and has the equation

$$x = x(t_0)$$

Mathematical details will be provided in class.

Examples

Example 21.1.1

Consider the cycloid

$$x(\theta) = 2(\theta - \sin\theta) \qquad y(\theta) = 2(1 - \cos\theta)$$

i) Find the equation of the tangent line at the point corresponding to $\theta = \frac{2\pi}{3}$ to this curve. Plot the cycloid and the tangent line in the same figure.

ii) Find the area of the region under one arc of the cycloid and above the x-axis.

Solution

i)

Code the parametric equations and create the equation of the tangent line

```
> f:=theta->2*(theta-sin(theta));
  g:=theta->2*(1-cos(theta));
```

$$f := \theta \to 2\theta - 2\sin(\theta)$$

$$g := \theta \to 2 - 2\cos(\theta) \qquad \textbf{(2.1.1.1)}$$

Find the slope of the tangent line.

```
> m:=D(g)(2*Pi/3)/D(f)(2*Pi/3);
```

$$m := \frac{1}{3}\sqrt{3} \qquad (2.1.1.2)$$

Code the equation of the tangent line slope-intercept form.

```
> TL:=simplify(y=g(2*Pi/3)+m*(x-f(2*Pi/3)));
```

$$TL := y = 4 + \frac{1}{3}\sqrt{3}\,x - \frac{4}{9}\sqrt{3}\,\pi \qquad (2.1.1.3)$$

Create the plot of the curve and the tangent line.

```
> p1:=plot([f(theta), g(theta), theta=-Pi..8*Pi], color=
  red):
  p2:=plot(rhs(TL), x=-Pi..2*Pi, color=blue):
> display([p1, p2], scaling=constrained, view=[-3..15, -1.
  .5], tickmarks=[15,5]);
```



ii) Find the area of the region under one arc of the cycloid and above the *x*-axis.

The area of the region enclosed by the *x*-axis and one arc of this cycloid is given by

$$\int_0^{2\pi} y(\theta)x'(\theta)\,d\theta$$

```
> e1:=Int(g(theta)*D(f)(theta), theta=0..2*Pi);
```

$$e1 := \int_0^{2\pi} (2 - 2\cos(\theta))^2\,d\theta \qquad (2.1.1.4)$$

```
> area:=value(e1);
```

$$area := 12\,\pi \qquad (2.1.1.5)$$

>